Лабораторная работа №2

АРИФМЕТИЧЕСКИЕ ВЫЧИСЛЕНИЯ НА АССЕМБЛЕРЕ 

1. СОДЕРЖАНИЕ РАБОТЫ
- изучение арифметических команд сложения, вычитания, умножения и деления;

- изучение команд сравнения и перехода;
- написание и отладка программ вычисления арифметических выражений;

- работа с отладчиком.

2. КРАТКИЕ СВЕДЕНИЯ ОБ ИСПОЛЬЗУЕМЫХ КОМАНДАХ

2.1. Флаги

Особую роль при выполнении арифметических операций играют флаги, с помощью которых фиксируются признаки полученного результата. Все флаги хранятся в регистре FLAGS. Вот некоторые из них. 

CF (Carry Flag) - флаг переноса. Содержит значение "переносов" (0 или 1) из старшего разряда результирующего операнда при арифметических операциях и логических операциях сдвига.

PF (Parity Flag) - флаг четности. Проверяет младшие восемь бит результата операции над данными. Нечетное число бит приводит к установке этого флага в 0, а четное - в 1. 

ZF (Zero Flag) - флаг нуля. Устанавливается в качестве результата арифметических команд и команд сравнения. Ненулевой результат приводит к установке нулевого значения этого флага, а нулевой - к установке единичного значения. Команды условного перехода JE и JZ проверяют этот флаг.

SF (Sign Flag) - флаг знака. Устанавливается в соответствии со знаком результата (старшего бита) после арифметических операций: положительный результат устанавливает 0, а отрицательный - 1. Команды условного перехода JG и JL проверяют этот флаг.

OF (Overflow Flag) - флаг переполнения. Фиксирует арифметическое переполнение, т.е. перенос в/из старшего (знакового) бита при знаковых арифметических операциях.


  2.2 Арифметические команды

Типы операндов при описании команд будут представлены в символическом виде:  m - операнд в памяти, r - регистровый операнд,  i - непосредственный операнд, a - адресный операнд. Число, записываемое после символа, означает: 8 - байт, 16 - слово, 32 - двойное слово. Например: r8 - байтовый регистр, m16 - слово в памяти и т.п.

 2.2.1. Сложение

ADD приемник,источник
Действие команды: приемник := приемник + источник
Типы операндов (приемник и источник должны иметь одинаковый размер):

приемник – m8,r8,m16,r16,m32,r32

источник - m8,r8,i8,m16,r16,i16,m32,r32,i32

Устанавливаемые флаги: CF, OF, ZF, SF, PF
Примеры:

	ADD  AL,100   ; AL:=AL+100

ADD  MEM,1Fh  ; MEM:=MEM+1Fh

ADD  CX,DX
 ; CX:=CX+DX
	MOV  AH,100  ;AH:=100

ADD  AH,100  ;AH:=AH+100=200, CF=0

ADD  AH,56   ;AH:=AH+56=0, CF=1


2.2.2. Вычитание

SUB приемник,источник
Действие команды: приемник := приемник - источник
Типы операндов (согласуются по размеру):

приемник – m8,r8,m16,r16,m32,r32

источник - m8,r8,i8,m16,r16,i16,m32,r32,i32

Устанавливаемые флаги: CF, OF, ZF, SF, PF
Примеры:

	SUB AL,100   ; AL:=AL-100

SUB MEM,1Fh  ; MEM:=MEM-1Fh

SUB CX,DX
 ; CX:=CX-DX
	MOV  AH,100  ;AH:=100

SUB  AH,100  ;AH:=AH-100=0, CF=0, ZF=1

SUB  AH,1    ;AH:=AH-1=255, CF=1, ZF=0


 2.2.3. Умножение беззнаковых чисел 

MUL операнд

Действие команды (определяется типом операнда): 

тип операнда - m8,r8:
AX := операнд * AL


тип операнда - m16,r16:
(DX:AX):= операнд * AX



тип операнда - m32,r32:
(EDX:EAX):= операнд * EAX



Устанавливаемые флаги: CF, OF, ZF, SF, PF (CF и OF устанавливаются в 1, если старшая половина результата ненулевая)

Примеры:

	MOV AL,20

MOV BL,7

MUL BL    ; AX:=BL*AL=140<255

          ; CF=OF=0
	MOV AL,20

MOV BL,13

MUL BL     ; AX:=BL*AL=260>255

           ; CF=OF=1


 2.2.4. Умножение чисел с учетом знака

IMUL операнд

Действие команды и атрибуты соответствуют команде MUL
 2.2.5. Деление беззнаковых чисел 

DIV операнд

Действие команды (определяется типом операнда), частное размещается в 1-м регистре, остаток (MOD) во 2-м:

тип операнда - m8,r8



AL := AX / операнд, AH := MOD (AX / операнд) 

тип операнда - m16,r16


AX := (DX:AX)/ операнд, DX := MOD (AX / операнд) 

тип операнда - m32,r32


EAX :=(EDX:EAX)/ операнд, EDX := MOD (AX / операнд) 

Операнд имеет размер в два раза меньше, чем делимое, размещаемое в регистрах в AX, DX:AX или EDX:EAX.

Устанавливаемые флаги: нет. Вызывает исключение типа “ошибка деления” при делении на ноль либо, когда частное слишком велико для размещения и AL/AX/EAX.

Примеры:


MOV BL=17


MOV AX,175


DIV BL
; AL:=AX/BL=10, AH:=MOD(AX/BL)=5

 2.2.6. Деление чисел с учетом знака

IDIV операнд

Действие команды и атрибуты соответствуют команде DIV

 2.3. Команды сравнения и передачи управления

Команды передачи управления служат для скачкообразного перехода на указанную команду, которая помечается символической меткой. Переход реализуется путем изменения указателя команд CS:IP/EIP в соответствии с адресом размещения помеченной команды. В зависимости от местоположения метки и заданных атрибутов различают следующие типы перехода:

- ближний (NEAR) переход - если метка описана в том же сегменте кода, что и команда передачи управления;

- дальний (FAR) переход - если метка описана в другом сегменте кода;

- короткий (SHORT) переход - если величина перехода (расстояние между командой передачи управления и меткой) может быть представлена однобайтовым знаковым числом (-128..127);

- длинный (LONG) переход - если величина перехода может быть представлена двухбайтовым знаковым числом (-32768..32767);

2.3.1. Безусловный переход

JMP [тип] метка

Действие команды: безусловный переход на выполнение команды, на которую указывает заданная метка (возможно задание типа перехода). По умолчанию выполняется ближний переход (длинный - вперед и короткий - назад).

Тип операнда-метки – a8,a16,a32,m16,m32,m48

Устанавливаемые флаги: в реальном режиме нет

Примеры:

	    JMP CYCLE  ;переход вперед


. . .

CYCLE: MOV AX,1Fh
	      . . .

NEXT: MOV AX,1Fh    


. . .

   JMP NEXT  ;переход назад


 2.3.2. Сравнение

CMP операнд1,операнд2
Действие команды: производит вычитание операнд1-операнд2 и формирует флаги (операнды не изменяются, результат никуда не заносится!) как в команде вычитания.

Типы операндов (согласуются по размерам):

операнд1 – m8,r8,m16,r16,m32,r32

операнд2 - m8,r8,i8,m16,r16,i16,m32,r32,i32

Устанавливаемые флаги: CF, OF, ZF, SF, PF
Примеры:


MOV AX,20



CMP AX,10    ;20-10=10 -> CF=0,OF=0,ZF=0,SF=0


CMP AX,20    ;20-20=0 -> CF=0,OF=0,ZF=1,SF=0


CMP AX,30    ;20-30=-10 -> CF=1,OF=0,ZF=0,SF=1

 2.3.3. Условный переход

Jxx метка

Действие команды: переход на выполнение команды, которой предшествует указанная метка, в случае, если верно определенное условие. В противном случае выполняется следующая команда. Проверка условия связана с анализом состояния флагов, устанавливаемых, обычно, по команде сравнения CMP (но не обязательно!). Для МП 8086 выполняется только короткий переход!

Тип операнда: метка – a8,a16,a32,m16,m32,m48

Устанавливаемые флаги: нет

В таблице представлены поддерживаемые типы типа команд условного перехода:

	Команда
	Наименование
	Условия перехода 

	При сравнении чисел без знака по команде CMP

	JA или JNBE
	Переход если больше
	CF=0 и ZF=0

	JAE или JNB
	Переход если больше или равно 
	CF=0

	JB или JNAE 
	Переход если меньше
	CF=1

	JBE или JNA
	Переход если меньше или равно   
	CF=1 или ZF=1

	При сравнении чисел со знаком по команде CMP

	JG или JNLE 
	Переход если больше   
	ZF=0 и SF=OF

	JGE или JNL  
	Переход если больше или равно 
	SF=OF

	JL или JNGE 
	Переход если меньше  
	SF <> OF

	JLE или JNG   
	Переход если меньше или равно
	ZF=1 и SF <> OF

	При сравнении любых чисел по команде CMP

	JE  
	Переход если равно 
	ZF=1

	JNE 
	Переход если не равно 
	ZF=0

	При прямой проверке флагов

	JZ  
	Переход если ноль  
	ZF=1

	JNZ 
	Переход если не ноль  
	ZF=0

	JC  
	Переход если есть перенос 
	CF=1

	JNC   
	Переход если нет переноса
	CF=0

	JO  
	Переход если переполнение
	OF=1

	JNO
	Переход если нет переполнения  
	OF=0

	JS  
	Переход если знак минус 
	SF=1

	JNS 
	Переход если знак плюс 
	SF=0

	JP  
	Переход если четный
	PF=1

	JNP 
	Переход если нечетный
	PF=0

	При проверке регистра CX/ECX

	JCXZ 
	Переход если CX=0
	CX=0

	JECXZ 
	Переход если ECX=0
	ECX=0


Примеры:


CMP X,30


JGE CONT


. . .

CONT: MOV AX,0

3. РАБОЧЕЕ ЗАДАНИЕ

3.1. Написать и отладить программу вычисления функции Y(X) (по вариантам, представленным в таблице) с учетом следующих требований:

· переменные X и Y распределяются в памяти и имеют тип - байт, значения X и Y - беззнаковые целые числа;

· при выполнении арифметических операций там, где это необходимо, производить контроль флага переноса (СF) с целью выявления ошибочного результата;

· арифметические ошибки фиксировать с помощью переменной MYERR, которой присваивается значение 1 при наличии ошибки, и 0 в противном случае;

· в шапку программы включить в виде комментария сведения об авторе, номер варианта и расчетную формулу.

ВАРИАНТЫ ЗАДАНИЯ

	1
	   (3X-5,  X<=60

Y= (100-X/7, 60<X<=100

   (X+54,   X>100
	11
	   (X/6-3, X<30

Y= (2X+80, 30<=X<=70

   (140+X, X>70
	21


	   (120-5X,  X<30

Y= ((X+140)/X,30<=X<95

   (X+84,   X>=95

	2
	   (8X-10,  X<30

Y= (120+X/4, 30<=X<=160

   (X+25,   X>160
	12
	   (-43+6X, X<=40

Y= (10-200/X, 40<X<=150

   (X+99,  X>150
	22


	   (1000/(X+20), X<100

Y= (-120+3X,100<X<=120

   (40+X, X>120

	3
	   (5X-150, X<50

Y= (2X+50, 50<=X<=100

   (X/15+245,  X>100
	13
	   (8-2X,   X<=10

Y= (100/X-5, 10<X<20

   (X+170,   X>=20
	23
	   (-14X+250,  X<=20

Y= (X+1000/X, 20<X<200

   (10+X,  X>=200

	4
	   (6X-70,  X<=40

Y= (160+X/8, 40<X<110

   (50+X,   X>=110
	14
	   (256-2X,  X<90

Y= (-X/50+10,90<=X<=160

   (X+70, X>160
	24
	   ((70-X)/5, X<80

Y= (3X-160, 80<=X<=100

   (X+124, X>100

	5
	   (X/5-1, X<=30

Y= (3X-80, 30<X<=80

   (150+X, X>80
	15
	   (200/X-7, X<=45

Y= (155-2X, 45<X<=75

   (X/15+245,  X>75
	25
	   (4X+48,  X<=70

Y= (3X-210, 70<X<=120

   (235+X/10, X>120

	6
	   (65+4X, X<50

Y= (10-200/X, 50<=X<200

   (X+15,  X>=200
	16
	   (-3X+90,  X<=40

Y= (1+120/X, 40<X<60

   (140+X,   X>=60
	26
	   (280/(X+1), X<=65

Y= (185-2X, 65<X<90

   (X+54,  X>=90

	7
	   (5X+8,  X<40

Y= (4X-111, 40<=X<=60

   (X/12+240, X>60
	17
	   (180/X-1, X<=30

Y= (-3X-5, 30<X<=50

   (180+X, X>50
	27
	   ((X-2)/2, X<40

Y= (7X-250, 40<=X<=70

   (12+X, X>70

	8
	   (50-2X,  X<=30

Y= (X/4-1, 30<X<160

   (X+21,   X>=160
	18
	   (100-2X, X<=80

Y= (-1+250/X,80<X<=200

   (X+35,  X>200
	28
	   (840/(X+30), X<=140

Y= (-120+2X,140<=X<180

   (55+X, X>=180

	9
	   (2X-15, X<=90

Y= (250-2X, 90<X<=100

   (X/15+245,  X>100
	19
	   (8X-37,  X<=10

Y= (X-5,  10<X<=20

   (200/X+104, X>20
	29
	   (2X-15, X<200

Y= (250-X, 200<=X<220

   (X/11+238, X>=220

	10
	   (5X-15,  X<=40

Y= (160+X/8, 40<X<=200

   (X+50,   X>200
	20


	   (5X-18,  X<50

Y= ((90+X)/7,50<=X<=90

   (X+150,   X>90
	30
	   (255-5X,  X<55

Y= (-X/15+10,55<=X<135

   (X+17,  X>=135


3.2. Произвести выполнение программы в отладчике при четырех различных значениях переменной X и заполнить таблицу:

	
	X1=
	X2=
	X3=
	X4=

	Y (hex)
	
	
	
	

	Y (dec)
	
	
	
	


X1, X2, X3 – допустимые значения переменной X для каждого из трех заданных диапазонов, не приводящие к арифметическим ошибкам;

X4 – значение X, приводящее к арифметической ошибке (установлению флага переноса) при вычислениях в одном из диапазонов;

3.3. Выполняя программу в отладчике в пошаговом режиме сначала при X=X1, а затем при X=X4, заполнить таблицу: 

	№ пп
	Команда
	AX
	…
	IP
	X
	Y
	MYERR
	Флаги

	
	
	
	
	
	
	
	
	СF
	OF
	ZF
	SF

	1
	
	
	
	
	
	
	
	
	
	
	

	2
	
	
	
	
	
	
	
	
	
	
	

	…
	
	
	
	
	
	
	
	
	
	
	


В таблицу включить все используемые в программе регистры и, кроме того, регистр IP, память в ячейках X, Y и MYERR, а также флаги CF, OF, ZF, SF.

4. ТРЕБОВАНИЯ К ОТЧЕТУ

- продемонстрировать  работу программы преподавателю;

- включить в отчет: расчетную формулу (вариант задания), блок-схему алгоритма программы, текст программы, заполненные таблицы (п.3.2 и п.3.3)

5.КОНТРОЛЬНЫЕ ВОПРОСЫ

1. Почему команды умножения и деления используют только один явный операнд?

2. Как согласуются типы операндов при выполнении различных арифметических операций?

3. Что означают и как формируются флаги переноса (CF), переполнения (OF), знака (SF) и нуля (ZF) при выполнении различных арифметических операций?

4. В чем сходство и различие операций вычитания (SUB) и сравнения (CMP)?

5. Почему используются разные команды условного перехода при сравнении знаковых и беззнаковых чисел?

6. Поясните различия между длинным, коротким, ближним и дальним переходами.

ПРИЛОЖЕНИЕ. ПРИМЕР ПРОГРАММЫ

;Лабораторная работа №2

;Выполнил ст. гр. УXX-211 XXXXXXXXXX YYYYYYY
;Вариант 1234

;     ┌ 220-8*x,  x<30

;  y= ┤ 800/x-10, 30<=x<75

;     └ x+145, x>=75

model small

.stack 100h
;=========================================

.data

x db 100

y db ?

myerr db 0

;=========================================

.code

main:

mov ax,@data


mov ds,ax

; формирование регистра сегмента данных ds
;---------------------------------

cmp x,30

; х-30 ?

jae int2

; если x>=30 перейти к метке int2

; вычисление 220-8*x
mov al,8



mul x

; ax:=8*x 



; дальше учитываем, что произведение целиком находится в al
mov bl,220


sub bl,al

; bl:=220-8*x

jc  osh

; если есть перенос - ошибка, перейти к метке osh
mov al,bl

; результат сохраняем в al
jmp fin

; перейти к метке fin
;--------------------------------

int2:   cmp x,75

; х-75 ?

jae int3

; если x>=75 перейти к метке int3

; вычисление 800/x-10

mov ax,800

div x

; al:=ax/x=800/x
sub al,10

; al:=800/x-10

jmp fin

; перейти к метке fin
;--------------------------------

; вычисление x+145
int3:   mov al,x

add al,145
; al:=x+145

jnc fin

; если флаг переноса не установлен перейти к fin
osh: mov myerr,1
; установить код ошибки

jmp exit

; перейти к метке exit
;--------------------------------

fin:    mov y,al

; сформировать результат в переменной y
exit:   mov al,myerr
; код завершения программы

  mov ah,4ch 

int 21h

end main
